

A LITTLE WALLPAPER FOR THE APPLE ///

By

David Ottalini

/// SIG Co-Chairman

Here's a Business Basic program by Pair Software's Frank Moore that will produce excellent Mandelbrot graphs. You will need the Bgraf.Inv invokable module for it to work (it's on some of our PD offerings if you don't already have it). You GS Basic users may also enjoy adapting this for use on the GS.

```
10 REM PROGRAM ** WALLPAPER **
20 ON ERR GOSUB 1000:END
30 INVOKE"bgraf.inv"
40 GOSUB 500
45 HOME:PRINT CHR$(4);
50 PRINT"This program is an adaption of A.K. Dewdney's article on"
51 PRINT"'Wallpaper for the Mind' in the Computer Recreations section"
52 PRINT"of the September 1986 issue of Scientific American."
55 PRINT
60 PRINT"The program draws very interesting, if not quite repetitive,"
61 PRINT"graphics images. If you love Mandelbrot graphs, you'll love"
62 PRINT"these graphs. Use CONTROL-C RETURN to end the program."
70 PRINT
80 PRINT"As per the article, we suggest that you start with values of"
85 PRINT"15, 20 and 87 for Corner A, Corner B and Side."
86 PRINT
90 PRINT"Frank W. Moore, 74666,3660";CHR$(15)
100 VPOS=14:INPUT"Input Corner A: ";cornas$
105 IF cornas$="" THEN PRINT CHR$(7);:GOTO 100
110 corna=VAL(cornas$)*-1
112 IF corna>1 OR corna<-100 THEN PRINT CHR$(7);:GOTO 100
115 VPOS=15:INPUT"Input Corner B: ";cornb$
120 IF cornb$="" THEN PRINT CHR$(7);:GOTO 120
125 cornb=VAL(cornb$)*-1
130 IF cornb>1 OR cornb<-100 THEN PRINT CHR$(7);:GOTO 120
140 VPOS=16:INPUT"Input Side: ";side$
142 IF side$="" THEN PRINT CHR$(7);:GOTO 140
145 side=VAL(side$)
```

```

150 VPOS=17:INPUT"input graphic Mode (0,1,2 or 3): ";mode$
151 IF mode$="" THEN PRINT CHR$(7);:GOTO 150
153 mode%=VAL(mode$)
155 IF mode%<0 OR mode%>3 THEN PRINT CHR$(7);:GOTO 150
160 IF mode%<>1 THEN 180
170 VPOS=18:INPUT"Pen Color: ";pen$
171 IF pen$="" THEN PRINT CHR$(7);:GOTO 170
172 pen%=VAL(pen$):IF pen%<0 OR pen%>15 THEN PRINT CHR$(7);:GOTO 170
175 VPOS=19:INPUT"Fill Color: ";fill$
176 IF fill$="" THEN PRINT CHR$(7);:GOTO 175
177 fill%=VAL(fill$):IF fill%<0 OR fill%>15 THEN PRINT CHR$(7);:GOTO 175
178 IF fill%=pen% THEN PRINT CHR$(7);:GOTO 175
180 IF mode%=0 THEN pen%=15:fill%=0:ver%=279
185 IF mode%=1 THEN ver%=279
190 IF mode%=2 THEN pen%=15:fill%=0:ver%=559
200 IF mode%=3 THEN fill%=0:ver%=139
210 VPOS=21:HPOS=1:PRINT"Do design in the ";modes$(mode%);
215 PRINT" x 192 grafix mode with pen and fill colors"
220 IF mode%=3 THEN PRINT"of various pen colors and black.":GOTO 240
230 PRINT"of ";pens$(pen%);" and ";fill$(fill%);"."
240 PRINT:PRINT"Press RETURN to begin, ESCAPE to start again...";
250 GET char$
252 IF char$=CHR$(27) THEN 45
255 IF char$<>CHR$(13) THEN 250
257 IF mode%<>3 THEN GOSUB 400
259 REM **** SET UP GRAPHICS MODE AND SCREEN ****
260 PERFORM grafixmode(%mode%,%1)
270 PERFORM grafixon
280 PERFORM fillcolor(%fill%)
290 PERFORM fillport
294 REM **** ON KEYPRESS RETURN TO MAIN MENU ****
295 ON KBD POP:GOSUB 1000:GOTO 50
298 REM **** FIND AND PRINT SCREEN PIXELS ****
299 REM **** THIS IS MAIN PART OF PROGRAM ****
300 FOR x%=zero% TO ver%:REM **** SET PIXEL COLUMN ****
310   FOR y%=zero% TO hor%:REM **** SET PIXEL ROW ****
320     i=corna+(side*x%/ver%)
330     j=cornb+(side*y%/hor%)
340     z=(i*i)+(j*j)
350     c%=INT(z)
359     REM **** IF MODE 3 THEN CALCULATE PEN COLOR AND SET ****
360     IF mode%=3 THEN pen%=(c%-(INT(c%/16)*16)):GOSUB 00:GOTO 380

```

```

368 REM **** OTHER MODES PRINT DOT IN CURRENT PEN COLOR ****
369 REM **** IF c% IS AN EVEN NUMBER ****
370 IF c%/two%<>INT(c%/two%) THEN 390
379 REM **** PRINT SCREEN DOT IN CURRENT PEN COLOR ****
380 PERFORM dotat(%x%,%y%)
390 NEXT
395 NEXT
396 REM **** IF GRAPH COMPLETE, SAVE AND RETURN TO MAIN MENU ****
397 PERFORM gsave."w.graph":GOSUB 1000:GOTO 45
399 REM **** SET PEN COLOR ****
400 PERFORM pencolor(%pen%)
410 RETURN
499 REM **** INITIALIZE VARIABLES ****
500 hor%=191:zero%=0:two%=2:DIM modes$(3),pens$(15),fill$(15)
510 modes$(0)="280":modes$(1)="280":modes$(2)="560":modes$(3)="1 40"
520 pens$(0)="Black":fill$(0)="Black"
530 pens$(1)="Magenta":fill$(1)="Magenta"
540 pens$(2)="Dark Blue":fill$(2)="Dark Blue"
550 pens$(3)="Purple":fill$(3)="Purple"
560 pens$(4)="Dark Green":fill$(4)="Dark Green"
570 pens$(5)="Gray 1":fill$(5)="Gray 1"
580 pens$(6)="Medium Blue":fill$(6)="Medium Blue"
590 pens$(7)="Light Blue":fill$(7)="Light Blue"
600 pens$(8)="Brown":fill$(8)="Brown"
610 pens$(9)="Orange":fill$(9)="Orange"
620 pens$(10)="Grey 2":fill$(10)="Grey 2"
630 pens$(11)="Pink":fill$(11)="Pink"
640 pens$(12)="Green":fill$(12)="Green"
650 pens$(13)="Yellow":fill$(13)="Yellow"
660 pens$(14)="Aqua":fill$(14)="Aqua"
670 pens$(15)="White":fill$(15)="White"
700 RETURN
999 REM **** HOME, RETURN TO TEXT SCREEN AND RELEASE GRAPHICS MEMORY ****
1000 OFF KBD:HOME:TEXT:PERFORM release:PERFORM release:PERFORM release
1010 RETURN

```